Fourier'n muunnos

Määritelmiä

Funktion f Fourier'n muunnos $F = \mathcal{F}f$ on

$$F(u) = \int_{-\infty}^{\infty} f(x)e^{-2\pi iux}\,dx.$$

ja käänteismuunnos

$$f(x) = \int_{-\infty}^{\infty} F(u)e^{2\pi iux}\,du.$$

Muitakin määritelmiä käytetään yleisesti. Integraalin edessä voi olla vakiokertoimia, kuten $1/2\pi$ tai $1/\sqrt{2\pi}$. Eksponentin tekijä 2π voi puuttua tai merkit voivat olla toisinpäin. Tarkista aina, mitä määritelmää milloinkin on käytetty!

Esimerkiksi yksikköpulssin $\operatorname{rect}(x)$ Fourier'n muunnos:

$$\begin{aligned}
R(u) &= \int_{-\infty}^{\infty} \operatorname{rect}(x)e^{-2\pi iux}dx \\
&= \int_{-1/2}^{1/2} e^{-2\pi iux}dx \\
&= \frac{1}{-2\pi iu}(e^{-\pi iu} - e^{\pi iu}) \\
&= \frac{1}{\pi u}\frac{e^{\pi iu} - e^{-\pi iu}}{2i} \\
&= \frac{\sin \pi u}{\pi u} = \operatorname{sinc}(\pi u),
\end{aligned}$$

Yksikköpulssin avulla voidaan määritellä funktiot

$$\delta_n(x) = n \operatorname{rect}(nx).$$

Funktion δ_n Fourier'n muunnos on $\operatorname{sinc}(\pi u/n)$.

Raja-arvona saadaan Diracin delta-funktio.

$$\delta(x) = \lim_{n\to\infty} \delta_n(x) = \begin{cases} 0 & \text{kun } x \neq 0 \\ \infty & \text{kun } x = 0 \end{cases}.$$

Fourier'n muunnoksen ominaisuuksia

$$\mathcal{F}(af + bg) = a\mathcal{F}f + b\mathcal{F}g.$$

$$\mathcal{F}[f(ax)] = \frac{1}{|a|}F(u/a).$$

$$\mathcal{F}[f(x-a)] = F(u)e^{-2\pi iua}.$$

Parsevalin teoreema:

$$\int_{-\infty}^{\infty} f(x)g^*(x)\,dx = \int_{-\infty}^{\infty} F(u)G^*(u)\,du,$$

missä tähti (*) tarkoittaa kompleksikonjugaattia.

Jos valitaan $g = f$, saadaan

$$\int_{-\infty}^{\infty} |f(x)|^2\,dx = \int_{-\infty}^{\infty} |F(u)|^2\,du.$$

Konvoluutio

Funktioiden h ja f konvoluutio on

$$(h * f)(x) = \int_{-\infty}^{\infty} h(x - x')f(x')\,dx'.$$

Konvoluution Fourier'n muunnos on funktioiden muunnosten tulo:

$$\mathcal{F}(h * f)(u) = (\mathcal{F}h)(u)(\mathcal{F}f)(u) = H(u)F(u).$$

Funktioiden f ja g korrelaatio R_{fg} on

$$R_{fg}(x) = \int_{-\infty}^{\infty} f(x')g(x + x')\,dx'.$$

Funktion korrelaatio itsensä kanssa on autokorrelaatiofunktio:

$$R_{ff}(x) = \int_{-\infty}^{\infty} f(x')f(x + x')\,dx',$$

jonka Fourier'n muunnos on

$$\mathcal{F}R_{ff} = |F(u)|^2.$$

Tehospektri (power spectral density, PSD):

$$P_f(u) = |F(u)|^2 + |F(-u)|^2.$$

Jos f on reaalinen, on

$$P_f(u) = 2|F(u)|^2.$$

Diskreetti Fourier'n muunnos

Oletetaan, että muunnettava funktio tunnetaan tasavälein sijaitsevissa pisteissä $f_n = f(n\Delta x)$, $n = 0, \dots, N-1$. Se voidaan siten esittää N:n alkion vektorina $\mathbf{f} = (f_0, \dots, f_{N-1})$.

Diskreetti Fourier'n muunnos määritellään:

$$F_k = \sum_{n=0}^{N-1} f_n \exp\left(\frac{-2\pi nik}{N}\right).$$

Käänteismuunnos on

$$f_n = \frac{1}{N}\sum_{k=0}^{N-1} F_k \exp\left(\frac{2\pi nik}{N}\right).$$

Todistus:

$$\begin{aligned}
&\frac{1}{N}\sum_{k=0}^{N-1} F_k \exp\left(\frac{2\pi nik}{N}\right) \\
=&\frac{1}{N}\sum_{k=0}^{N-1}\left(\sum_{n'=0}^{N-1} f_{n'} \exp\left(\frac{-2\pi n'ik}{N}\right)\right)\exp\left(\frac{2\pi nik}{N}\right) \\
=&\frac{1}{N}\sum_{n'=0}^{N-1}\left(f_{n'}\sum_{k=0}^{N-1}\exp\left(2\pi i(n-n')\frac{k}{N}\right)\right).
\end{aligned}$$

Tässä esiintyvä sisempi summa on geometrinen sarja, jonka arvo on

$$\sum_{k=0}^{N-1}\exp\left(2\pi i(n-n')\frac{k}{N}\right) = \begin{cases} N, & \text{kun } n = n', \\ 0, & \text{muuten.}\end{cases}$$

Siten koko lausekkeen arvo on

$$\frac{1}{N}\sum_{n'=0}^{N-1} f_{n'} N\delta_{nn'} = \frac{1}{N} f_n N = f_n.$$

Normitustekijä $1/N$ esiintyy vain jommassakummassa muunnoksessa. Joskus sitä käytetään suoran muunnoksen lausekkeessa, jolloin se ei esiinny käänteismuunnoksessa.

Diskreetissä tapauksessa funktion ja sen muunnoksen arvoja esittävät luvut f_n ja F_k liittyvät itse funktioiden f ja F arvoihin kaavoilla

$$f_n = \Delta x f(n\Delta x),$$
$$F_k = F(k\Delta u),$$

missä

$$\Delta u \Delta x = \frac{1}{N}.$$

Kaksiulotteisen kuvan muunnos saadaan vastaavalla tavalla:

$$F_{k,l} = \sum_{n=0}^{N-1} \sum_{m=0}^{M-1} f_{n,m} \exp\left(-2\pi i \left(\frac{nk}{N} + \frac{ml}{M}\right)\right),$$

Käänteismuunnos on

$$f_{n,m} = \frac{1}{NM} \sum_{k=0}^{N-1} \sum_{l=0}^{M-1} F_{k,l} \exp\left(2\pi i \left(\frac{nk}{N} + \frac{ml}{M}\right)\right),$$

Näytteenottolause ja Nyquistin taajuus

Millä edellytyksillä alkuperäinen signaali voidaan rekonstruoida diskreeteistä näytteistä?

Tunnetaan funktion $f = f(x)$ arvot tasavälisissä pisteissä $x = kT$. Muodostetaan funktio g siten, että

$$f(x) = \sum_{k=-\infty}^{\infty} f(kT)g(x - kT).$$

Oletetaan, että funktioilla f ja g on Fourier'n muunnokset.

$$\begin{aligned} f(x) &= \sum_{k=-\infty}^{\infty} \int_{-\infty}^{\infty} f(t)g(x-t)\delta(t-kT)\,dt \\ &= \int_{-\infty}^{\infty} f(t)g(x-t)\left(\sum_{k=-\infty}^{\infty} \delta(t-kT)\right) dt. \end{aligned}$$

Suluissa oleva olio koostuu äärettömän monesta piikistä, jotka sijaitsevat etäisyydellä T toisistaan. Jaksollisuutensa vuoksi se voidaan esittää Fourier'n sarjana:

$$\sum_{k=-\infty}^{\infty} \delta(t-kT) = \sum_{n=-\infty}^{\infty} a_n e^{(2\pi int/T)},$$

missä

$$\begin{aligned} a_n &= \frac{1}{T}\int_{-T/2}^{T/2} \left(\sum_{k=-\infty}^{\infty} \delta(t-kT)\right) e^{-2\pi int/T}\,dt \\ &= \frac{1}{T}\int_{-T/2}^{T/2} \delta(t)e^{-2\pi int/T}\,dt \\ &= \frac{1}{T}. \end{aligned}$$

Niin ollen on

$$f(x) = \sum_{n=-\infty}^{\infty} \int_{-\infty}^{\infty} \left(f(t)e^{2\pi int/T}\right)\left(\frac{g(x-t)}{T}\right) dt.$$

Tämä on konvoluutioiden summa. Olkoot $F(u)$ ja $G(u)$ funktioiden f ja g Fourier'n muunnokset. Funktioiden

$$f(t)e^{2\pi int/T} \quad \text{ja} \quad \frac{g(x-t)}{T}$$

muunnokset ovat

$$F(u - n/T) \quad \text{ja} \quad G(u)/T.$$

Konvoluutiolauseen avulla saadaan

$$F(u) = \frac{G(u)}{T} \sum_{n=-\infty}^{\infty} F(u - n/T).$$

Oletetaan nyt, että $F(u) = 0$, kun $|u| \geq f_c$ jollekin f_c. Tämä tarkoittaa, että f ei sisällä tätä rajaa korkeampia taajuuksia. $F(u - n/T)$ on funktion $F(u)$ kopio, jota on siirretty askelen n/T verran. Jos $T \leq 1/2f_c$, nämä funktion F kopiot eivät ulotu toistensa päälle. Jos valitaan

$$G(u) = \begin{cases} T, & \text{jos } |u| < f_c, \\ 0, & \text{muuten,} \end{cases}$$

yhtälö toteutuu.

Nyt g saadaan ottamalla funktion G käänteismuunnos:

$$\begin{aligned} g(x) &= \int_{-f_c}^{f_c} T e^{i2\pi ux}\, du \\ &= T \frac{\sin 2\pi f_c x}{\pi x} \\ &= 2T f_c \operatorname{sinc}(2\pi f_c x). \end{aligned}$$

Jos taas $T > 1/2f_c$, F:n kopiot menevät päällekkäin. Silloin emme pysty löytämään sellaista funktiota G, joka toteuttaisi yhtälömme.

Näytteenottolause (sampling theorem):

Jos funktion f Fourier'n muunnos häviää kaikille $|u| \geq f_c$, f voidaan rekonstruoida täysin f:n arvoista pisteissä, joiden etäisyys on $1/2f_c$ tai pienempi.

Jos signaali sisältää taajuuksia, jotka ovat suurempia kuin puolet näytteidenottotaajuudesta, signaalia ei voi enää rekonstruoida yksikäsitteisellä tavalla. Nämä korkeat taajuudet saattavat aiheuttaa alias-ilmiön (aliasing), eli rekonstruoituun signaaliin ilmaantuu virheellisen tulkinnan vuoksi taajuuksia, joita alkuperäisessä signaalissa ei ole.

FFT (Fast Fourier Transform)

Menetelmä liitetään yleensä Cooleyn ja Tukeyn julkaisuun vuodelta 1965, vaikka vastaavanlaisia ajatuksia tunnettiin paljon aiemminkin (Gauss 1805). Parhaimmillaan menetelmä on, kun muunnettavan vektorin pituus on kakkosen potenssi, mutta algoritmista on olemassa versioita myös yleiselle tapaukselle.

Kun munnettavan vektorin alkioiden määrä N on kakkosen potenssi, muunnos voidaan laskea ajassa $N \log N$.

Danielsonin ja Lanczosin lemma(1942):

$$
\begin{aligned}
F_k &= \sum_{j=0}^{N-1} f_j e^{-2\pi ijk/N} \\
&= \sum_{j=0}^{N/2-1} f_{2j} e^{-2\pi i(2j)k/N} + \sum_{j=0}^{N/2-1} f_{2j+1} e^{-2\pi i(2j+1)k/N} \\
&= \sum_{j=0}^{N/2-1} f_{2j} e^{-2\pi ijk/(N/2)} + W^k \sum_{j=0}^{N/2-1} f_{2j+1} e^{-2\pi ijk/(N/2)} \\
&= F_k^{\text{parillinen}} + W^k F_k^{\text{pariton}},
\end{aligned}
$$

missä

$$W = e^{-2\pi i/N}$$

Lasketaan siis erikseen parillisten ja parittomien alkioiden muunnokset, ja niistä yhdistetään lopullinen muunnos.

Kumpikin osa voidaan edelleen käsitellä samalla tavalla. Jatketaan rekursiivisesti, kunnes muunnettavana on vain yksi alkio.

```
program fft
 implicit none
 real, parameter :: pi=3.141592654
 integer, parameter :: m=2, n=2**m
 complex a(0:n)
 integer i

 a(0)=(0.0, 0.0) ;  a(1)=(1.0, 0.0)
 a(2)=(2.0, 0.0) ;  a(3)=(3.0, 0.0)

 write (*,*) 'alkuperainen data'
 do i=0,n-1
  write(*,'(2F8.4)') a(i)
 end do

 ! ensin suora muunnos
 call fourier(a, m, n, 1)
 write (*,*) 'suora muunnos'
 do i=0,n-1
  write(*,'(2F8.4)') a(i)
 end do

 ! sitten kaanteismuunnos
 call fourier(a, m, n, -1)
 write (*,*) 'kaanteismuunnos '
 do i=0,n-1
  write(*,'(2F8.4)') a(i)
 end do

 contains

subroutine fourier (a, m, n, sig)
!  kompleksimuuttujan Fourier'n muunnos
!  alkuperainen data on vektorissa a(0:n-1)
!  muunnettavan vektorin pituuden n on oltava
!  kakkosen potenssi, n=2**m
!  muunnos korvaa alkuperaisen vektorin a(0..n-1)
!  jos sig < 0, lasketaan kaanteinen muunnos
 implicit none
 integer, intent(in) :: m, n, sig
 complex, dimension(0:n), intent(inout) :: a
 integer i,j,k,l, half_n, le,le1,ip
 real  angle
 complex x,u,w,t

 half_n = n/2

 ! jos kaanteismuunnos, konjugoidaan a
 if (sig < 0) a = conjg(a)

 ! jarjestellaan vektori a s.e. a(s)=a(j), missa
 ! s on j:n binaariesitys takaperin
 j=1
 do i=1, n-1
  if (i < j)  then
```

```
    x = a(j-1) ; a(j-1) = a(i-1) ;  a(i-1) = x ;
   end if
   k = half_n
   do while (k < j)
     j = j-k ;  k = k/2
   end do
   j = j+k
  end do

  ! muunnetaan palasia, joiden pituus on
  ! le=2,4,8,...
  le=2
  do l=1, m
   le1 = le/2
   u = cmplx(1,0)
   angle = pi/le1
   w = cmplx(cos(angle),sin(angle))
   do j=1,le1
    do i=j,n,le
     ip=i+le1
     x = a(ip-1)
t = x * u
a(ip-1) = a(i-1)-t
a(i-1) = a(i-1)+t
    end do
    x = u * w
    u = x
   end do
   le = 2*le
  end do

  ! jos kaanteismuunnos, konjugoidaan a takaisin
  ! ja normitetaan
  if (sig < 0) a=conjg(a)/n
end subroutine
end program
```

Ohjelman tulostus:

```
alkuperainen data
 0.0000  0.0000
 1.0000  0.0000
 2.0000  0.0000
 3.0000  0.0000
suora muunnos
 6.0000  0.0000
-2.0000 -2.0000
-2.0000  0.0000
-2.0000  2.0000
kaanteismuunnos
 0.0000  0.0000
 1.0000  0.0000
 2.0000  0.0000
 3.0000  0.0000
```

Äärellinen ja ei-tasavälinen aineisto

Deeming: Fourier analysis with unequally-spaced data, *Astrophysics and Space Science* **36** 137–158 (1975).

Dutt, Rokhlin: Fast Fourier transforms for nonequispaced data, *SIAM J. Sci. Comput.* **14**, No 6, 1368–1393 (1993)

Äärellisen aineiston muunnos

$$F_T(u) = \int_{-T/2}^{T/2} f(t) e^{-2\pi i u t}\, dt$$

voidaan kirjoittaa muotoon

$$F_T(u) = \int_{-\infty}^{\infty} w_T f(t) e^{-2\pi i u t}\, dt = \mathcal{F}(w_T f),$$

missä w_T on dataikkuna

$$w_T(t) = \begin{cases} 1, & \text{kun } t \in [-T/2, +T/2], \\ 0 & \text{muuten} \end{cases}$$

Diskreetti muunnos saadaan käyttämällä ikkunaa

$$w_N(t) = \sum_{n=0}^{N-1} \delta(t - t_n),$$

jolloin

$$\begin{aligned} F_N(u) &= \int_{-\infty}^{\infty} \sum_n \delta(t - t_n) f(t) e^{-2\pi i u t}\, dt \\ &= \sum_n f(t_n) e^{-2\pi i u t_n}. \end{aligned}$$

Tähtitieteessä havainnot eivät juuri koskaan ole tasavälisiä (esim. tähden valokäyrä). Ikkunan w_N lausekkeessa ajat t_k voivat olla mielivaltaisia, joten diskreetti muunnos voidaan laskea oleellisesti samalla tavalla myös ei-tasaväliselle aineistolle.